Dangling Stick

This *Mathematica* notebook derives the equations of motion for a Dangling Stick. We have a stick with a point mass at each end, and one end is attached to a spring whose other end is fixed. The system is free to move in 2 dimensions.

For more information and a JavaScript simulation, see the webpage
`www.myphysicslab.com/dangle_stick.html`.

Author: Erik Neumann, Jan 7, 2002

Kinematics

R1 is the position of the mass at spring-stick intersection.
R2 is the position of the mass at free end of stick.

```
R1[θ_, ϕ_, r_] := {r Sin[θ], -r Cos[θ]}
```

```
R2[θ_, ϕ_, r_] := {r Sin[θ] + L Sin[ϕ], -(r Cos[θ] + L Cos[ϕ])}
```

```
R1[0, 0, 1]
```

```
{0, -1}
```

```
R2[0, 0, 1]
```

```
{0, -1 - L}
```

Constants:
m1, m2 are masses
L is length of stick
r is length of spring
k is spring constant
h is spring rest length
g is gravity

```
SetAttributes[{L, m1, m2, k, h, g}, Constant]
```

```
Dt[R1[θ, ϕ, r]]
```

```
{r Cos[θ] Dt[θ] + Dt[r] Sin[θ], -Cos[θ] Dt[r] + r Dt[θ] Sin[θ]}
```

Velocities

Here we derive the velocities of the two masses. Note that we replace the derivatives with new variables... this is necessary to carry out the Lagrangian differentiation.

```
v1 = Dt[R1[θ, ϕ, r]] /. {Dt[θ] → θ1, Dt[ϕ] → ϕ1, Dt[r] → r1}
```

```
{r θ1 Cos[θ] + r1 Sin[θ], -r1 Cos[θ] + r θ1 Sin[θ]}
```

```
v2 = Dt[R2[θ, ϕ, r]] /. {Dt[θ] → θ1, Dt[ϕ] → ϕ1, Dt[r] → r1}
```

```
{r θ1 Cos[θ] + L ϕ1 Cos[ϕ] + r1 Sin[θ], -r1 Cos[θ] + r θ1 Sin[θ] + L ϕ1 Sin[ϕ]}
```

Energy & Lagrangian

The coenergy.

```
T = m1/2 v1.v1 + m2/2 v2.v2 // Expand // Simplify
```

$$\frac{1}{2}\left(m1\, r1^2 + m2\, r1^2 + m1\, r^2\, \theta1^2 + m2\, r^2\, \theta1^2 + L^2\, m2\, \phi1^2 + 2\, L\, m2\, r\, \theta1\, \phi1\, \text{Cos}[\theta - \phi] + 2\, L\, m2\, r1\, \phi1\, \text{Sin}[\theta - \phi]\right)$$

The potential energy

```
V = k/2 (r - h)^2 - m2 g (r Cos[θ] + L Cos[ϕ]) - m1 g r Cos[θ]
```

$$\frac{1}{2}\, k\, (-h + r)^2 - g\, m1\, r\, \text{Cos}[\theta] - g\, m2\, (r\, \text{Cos}[\theta] + L\, \text{Cos}[\phi])$$

The Lagrangian.

```
ℒ = T - V
```

$$-\frac{1}{2}\, k\, (-h + r)^2 + g\, m1\, r\, \text{Cos}[\theta] + g\, m2\, (r\, \text{Cos}[\theta] + L\, \text{Cos}[\phi]) +$$
$$\frac{1}{2}\left(m1\, r1^2 + m2\, r1^2 + m1\, r^2\, \theta1^2 + m2\, r^2\, \theta1^2 + L^2\, m2\, \phi1^2 + 2\, L\, m2\, r\, \theta1\, \phi1\, \text{Cos}[\theta - \phi] + 2\, L\, m2\, r1\, \phi1\, \text{Sin}[\theta - \phi]\right)$$

Lagrangian Equations

Calculate the Lagrangian equation for each of the three variables.

```
eqn1 = Dt[D[ℒ, θ1]] - D[ℒ, θ] /.
    {Dt[θ] → θ1, Dt[ϕ] → ϕ1, Dt[r] → r1, Dt[θ1] → θ2, Dt[ϕ1] → ϕ2} // Simplify
```

$$r\left(2\, m1\, r1\, \theta1 + 2\, m2\, r1\, \theta1 + m1\, r\, \theta2 + m2\, r\, \theta2 + L\, m2\, \phi2\, \text{Cos}[\theta - \phi] + g\, (m1 + m2)\, \text{Sin}[\theta] + L\, m2\, \phi1^2\, \text{Sin}[\theta - \phi]\right)$$

```
eqn2 = Dt[D[ℒ, r1]] - D[ℒ, r] /.
    {Dt[θ] → θ1, Dt[ϕ] → ϕ1, Dt[r] → r1, Dt[θ1] → θ2, Dt[ϕ1] → ϕ2, Dt[r1] → r2} // Simplify
```

$$-h\, k + k\, r + m1\, r2 + m2\, r2 - m1\, r\, \theta1^2 - m2\, r\, \theta1^2 - g\, (m1 + m2)\, \text{Cos}[\theta] - L\, m2\, \phi1^2\, \text{Cos}[\theta - \phi] + L\, m2\, \phi2\, \text{Sin}[\theta - \phi]$$

```
eqn3 = Dt[D[ℒ, ϕ1]] - D[ℒ, ϕ] /.
    {Dt[θ] → θ1, Dt[ϕ] → ϕ1, Dt[r] → r1, Dt[θ1] → θ2, Dt[ϕ1] → ϕ2, Dt[r1] → r2} // Simplify
```

$$L\, m2\left(L\, \phi2 + (2\, r1\, \theta1 + r\, \theta2)\, \text{Cos}[\theta - \phi] + \left(r2 - r\, \theta1^2\right) \text{Sin}[\theta - \phi] + g\, \text{Sin}[\phi]\right)$$

Equations of Motion

Solve for the second derivatives.

```
soln = Solve[{eqn1 == 0, eqn2 == 0, eqn3 == 0}, {θ2, r2, ϕ2}] // Simplify
```

$$\left\{\left\{\theta 2 \to \left(-2\, g\, m1\, (m1+m2)\, \mathrm{Sin}[\theta] - 2\, m1\, \left(2\, (m1+m2)\, r1\, \theta 1 + L\, m2\, \phi 1^2\, \mathrm{Sin}[\theta-\phi]\right) + k\, m2\, (h-r)\, \mathrm{Sin}[2\, (\theta-\phi)]\right) \Big/ (2\, m1\, (m1+m2)\, r),\right.\right.$$

$$r2 \to \frac{1}{2\, m1\, (m1+m2)} \left(2\, h\, k\, m1 + h\, k\, m2 - 2\, k\, m1\, r - k\, m2\, r + 2\, m1^2\, r\, \theta 1^2 + 2\, m1\, m2\, r\, \theta 1^2 + 2\, g\, m1\, (m1+m2)\, \mathrm{Cos}[\theta] + 2\, L\, m1\, m2\, \phi 1^2\, \mathrm{Cos}[\theta-\phi] - k\, m2\, (h-r)\, \mathrm{Cos}[2\, (\theta-\phi)]\right),\ \left.\left.\phi 2 \to -\frac{k\, (h-r)\, \mathrm{Sin}[\theta-\phi]}{L\, m1}\right\}\right\}$$